
FINISH 
/CLEAR 
C***************************************************************************

********* 
C*** 
C*** RUOTA ANALIZZATA CON ELEMENTI DI FOURIER 
C*** 
C*** PARAMETRI 
C*** 
/PREP7 
DA=90          ! diametro albero 
DM=130          ! diametro mozzo 
DB=350           ! diametro anello esterno 
DR=400           ! daiametro ruota 
LA= 200          ! lunghezza albero 
LM=130          ! lunghezza mozzo 
LR=120           ! lunghezza anello esterno 
RR=15             ! raggio raccordo 
SC=50             ! spessore cartella 
FR=20000           ! carico 
*ASK,NFOU,N° termini nello sviluppo in serie di Fourier : ,25 
C*** 
C*** KEYPONTS 
C*** 
K,1, 
K,2,DA/2 
K,3,DA/2,LA/2 
K,4,0,LA/2 
K,5,DR/2 
K,6,DR/2,LR/2 
K,7,DB/2,LR/2 
K,8,DB/2,SC/2 
K,9,DM/2,SC/2 
K,10,DM/2,LM/2 
K,11,DA/2,LM/2 
C*** 
C*** LINEE 
C*** 
L,1,2 
L,2,11 
L,11,3 
L,3,4 
L,4,1 
L,2,5 
L,5,6 
L,6,7 
L,7,8 
L,8,9 
L,9,10 
L,10,11 
C*** 
C*** RACCORDI 
C*** 
LSEL,,LOC,X,DM/2+0.1,DB/2-0.1 
LSEL,R,LOC,Y,0.1,SC/2+0.1 
*GET,NL1,LINE,,NUM,MAX 
LSEL,,LOC,X,DB/2-0.1,DB/2+0.1 
LSEL,R,LOC,Y,SC/2-0.1,LR/2+0.1 
*GET,NL2,LINE,,NUM,MAX 
LSEL,ALL 
LFILLT,NL1,NL2,RR 
LSEL,,LOC,X,DM/2-0.1,DM/2+0.1 
LSEL,R,LOC,Y,SC/2-0.1,LM/2+0.1 



*GET,NL2,LINE,,NUM,MAX 
LSEL,ALL 
LFILLT,NL1,NL2,RR 
C*** 
C*** AREE 
C*** 
LSEL,,LOC,X,DA/2-0.1,99999 
LSEL,R,LOC,Y,-0.1,LM/2+0.01 
AL,ALL 
LSEL,,LOC,X,-0.1,DA/2+0.01 
AL,ALL 
C*** 
C*** ELEMENTI 
C*** 
ET,1,25 
SMRTSIZE,3 
AMESH,ALL 
C*** 
C*** VINCOLI 
C*** 
LSEL,,LOC,Y,-0.1,0.01 
DL,ALL,,UY,0 
LSEL,ALL 
LSEL,,LOC,X,-0.01,0.01 
DL,ALL,,UX,0 
LSEL,ALL 
DK,2,UZ,0 
C*** 
C*** CARICHI 
C*** 
MP,EX,1,210000 
FINISH 
/SOLU 
*DO,IJK,1,NFOU,1 
SFLDELE,ALL,ALL 
C*** DEFINIZIONE ARMONICA 
MODE,IJK-1,1 
LSEL,,LOC,X,DR/2-0.01,DR/2+0.1 
*IF,IJK,LE,1,THEN 
           SFL,ALL,PRES,FR/LR/(2*3.14157*DR/2) 
*ELSE 
           SFL,ALL,PRES,FR/LR/(3.14157*DR/2) 
*ENDIF 
LSEL,ALL 
C*** SOLUZIONE 
SOLVE 
*ENDDO 
FINISH 
C*** 
C*** POST-PROCESSING 
C*** 
/POST1 
LCSUM,ALL 
C*** 
C*** OPZIONI DI RAPPRESENTAZIONE 
C*** 
C*** scelta oggetti da rappresentare 
/PBC,ALL,,1 
/PBC,NFOR,,0 
/PBC,NMOM,,0 
/PBC,RFOR,,0 
/PBC,PATH,,0 
/PNUM,NODE,0 



/PSF,PRESS,NORM,2 
C*** scelta disposizione finestra/legenda 
/PLOPTS,INFO,2 
C*** scelta colori 
/COLOR,PBAK,OFF 
C*** rappresenta la struttura indeformata 
/DSCALE,,OFF 
C*** 
C*** COMBINAZIONE CASI DI CARICO PER OTTENERE IL RISULTATO FINALE 
C*** 
*DO,IJK,1,NFOU,1 
           SET,IJK 
           LCWRITE,IJK 
*ENDDO 
LCASE,1 
*DO,IJK,2,NFOU,1 
           ESEL,ALL 
           LCOPER,ADD,IJK 
           NSEL,,LOC,X,(DA+DM)/4,(DB+DR)/4 
           ESLN,,1 
           /TITLE, ARMONICHE DA 1 A %IJK% 
           PLNSTR,S,X 
          *ASK,IFL,Premere un tasto per continuare,0 
           PLNSTR,S,EQV 
          *ASK,IFL,Premere un tasto per continuare,0 
           NSEL,,LOC,X,DB/2-RR-0.001,DB/2-RR+0.001 
           NSEL,R,LOC,Y,SC/2-0.001,SC/2+0.001 
           NSEL,R,LOC,Z,-0.0001,0.0001 
           /OUTPUT,RUOTAFOURIER,TXT,,APPEND 
           A=IJK 
           PRNSOL,S,COMP 
           PRNSOL,S,PRIN 
           /OUTPUT,TERM 
           NSEL,ALL 
           ESEL,ALL 
*ENDDO 
 
 
 
 


